
Tall-and-Skinny and Short-and-Wide Optimizations for QR and LQ Decompositions 

Intel® Math Kernel Library (Intel® MKL) 2017 updates 3 and later versions provide optimized 

functionality for calculating QR decompositions of tall-and-skinny (TS) matrices, and for calculating LQ 

decompositions of short-and-wide (SW) matrices. 

Several LAPACK algorithms depend on QR decompositions and directly benefit from optimizations of 

these routines for TS sizes. In addition, QR decompositions are utilized in many common numerical 

methods. Speedup of the QR/LQ calculation for TS/SW matrices therefore provides direct performance 

benefits to users of LAPACK routines, as well as far-reaching performance benefits to those utilizing 

LAPACK routines in QR/LQ-dependent numerical methods. 

A general description of the parallel TSQR and SWLQ algorithm, applied to a matrix 𝐴, follows (for a 

more detailed description, as well as a description of the sequential TSQR algorithm, please reference: 

http://www.netlib.org/lapack/lawnspdf/lawn204.pdf). The matrix 𝐴 is split along its largest axis into 

sub-matrices (for this example, 𝐴 is split into 4 submatrices), 

𝐴 =  [

𝐴0

𝐴1

𝐴2

𝐴3

]. 

The QR decompositions of these sub-matrices are then found, 

𝐴0 = 𝑄00𝑅00, 𝐴1 = 𝑄10𝑅10, 𝐴2 = 𝑄20𝑅20, 𝐴3 = 𝑄30𝑅30.  

The remaining 𝑅 factors can be consolidated vertically, and QR decompositions of these consolidated 

factors found to be 

[
𝑅00

𝑅10
] = 𝑄01𝑅01, [

𝑅20

𝑅30
] = 𝑄11𝑅11. 

This process repeats until all of the submatrices have been consolidated, at which point a final QR 

factorization will result in the following decomposition, 

𝐴 =  [

𝑄00    
 𝑄10   
  𝑄20  
   𝑄30

] ∙ [
𝑄01  

 𝑄11
] ∙ 𝑄02𝑅02, 

where the connection to the QR factorization is clear from this step. The SWLQ optimization is 

analogous, and thus will not be discussed in this article. 

The above description is just a general overview of the concepts behind the TSQR algorithm. In LAPACK, 

the TSQR calculation is actually performed using extra sequential QR calculations at each step, but these 

details are beyond the scope of this article. 

New routines have been added to Intel MKL to allow for the calculations of QR and LQ factorizations 

using the TS/SW modifications described above for appropriate matrix sizes. These routines are 

generalized for all sizes (i.e. they will also work on matrices that are not TS/SW, as they include paths to 

return to the generic routines when the matrix size is not sufficiently TS/SW). 

http://www.netlib.org/lapack/lawnspdf/lawn204.pdf


To find the complete QR decomposition of A, two steps are needed: calculation of A=QR, where Q is 

stored implicitly, and calculation of the explicit form of Q. Because the implicit representations of Q are 

not the same in the TS/SW optimized routines and the generic routines, the routines for finding A=QR 

and for finding the explicit representation of Q cannot be mixed TS/SW and generic, despite the fact that 

the TS/SW routines are generalized for all matrix sizes. 

Details of the new routines and parameter specifications can be found in the Intel MKL Developer 

Reference (https://software.intel.com/en-us/articles/intel-math-kernel-library-documentation). The 

routines to reference are listed below: 

New TS/SW Routine Generic Routines 

QR Decomposition 

 ?geqr 

 ?gemqr 
 
 
LQ Decomposition 

 ?gelq 

 ?gemlq 

QR Decomposition 

 ?geqrf 

 ?ormqr (real) 

 ?unmqr (complex) 
 
LQ Decomposition 

 ?gelqf 

 ?ormlq (real) 

 ?unmlq (complex) 
Table 1: TS/SW-optimized routines and generic routines 

An example of the typical workspace query/calculation calls for finding the QR decomposition of a 

double precision, TS matrix A is presented below. The first two lines call the routines to calculate the 

optimal workspace for the routine. The user is free to supply a smaller or larger workspace. If not 

https://software.intel.com/en-us/articles/intel-math-kernel-library-documentation


enough workspace is supplied for the calculation, info will be returned as the index of the insufficient 

workspace parameter.  

      INTEGER            M, N, K, LDA, LDC 

      PARAMETER          ( M = 1000, N = 5, K = 5, LDA = M, LDC = M ) 

      INTEGER            TSIZE, LWORK_QR, LWORK_MQR, INFO, ISEED( 4 ) 

* 

      DOUBLE PRECISION   A( LDA, N ), C( LDC, K ), 

     $                   TQUERY( 5 ), WQUERY( 1 ) 

      DOUBLE PRECISION, ALLOCATABLE :: T( : ), WORK_QR( : ), 

     $                   WORK_MQR( :) 

* 

      EXTERNAL           DGEQR, DGEMQR, DLARNV 

      INTRINSIC          INT, MAX 

* 

      DATA               ISEED / 0, 1, 2, 3 / 

* 

*     Initialize A as a random matrix 

      CALL DLARNV( 1, ISEED, M*N, A ) 

* 

*     Set C as the identity matrix 

      CALL DLASET( 'A', M, K, 0.0, 1.0, C, LDC ) 

* 

*     Workspace query for DGEQR 

      CALL DGEQR( M, N, A, LDA, TQUERY, -1, WQUERY, -1, INFO ) 

* 

*     Set size of T and workspace size for DGEQR, allocate T and WORK_QR 

      TSIZE = INT( TQUERY( 1 ) ) 

      LWORK_QR = INT( WQUERY( 1 ) ) 

      ALLOCATE( T( TSIZE ), WORK_QR( LWORK_QR ) ) 

* 

*     Find A=QR, store Q implicitly in A 

      CALL DGEQR( M, N, A, LDA, T, TSIZE, WORK_QR, LWORK_QR, INFO ) 

* 

*     Workspace query for DGEMQR, note that only the first 5 elements of 

*     T are used in the query call 

      CALL DGEMQR( 'L', 'N', M, K, N, A, LDA, T, TSIZE, C, LDC, 

     $             WQUERY, -1, INFO ) 

* 

*     Set size of workspace size and allocate WORK_MQR for DGEMQR 

      LWORK_MQR = INT( WQUERY( 1 ) ) 

      ALLOCATE( WORK_MQR( LWORK_MQR ) ) 

* 

*     Find Q explicitly and store in C 

      CALL DGEMQR( 'L', 'N', M, K, N, A, LDA, T, TSIZE, C, LDC, 

     $             WORK_MQR, LWORK_MQR, INFO ) 

* 

*     Deallocate memory for T, WORK_QR and WORK_MQR 

      DEALLOCATE( T, WORK_QR, WORK_MQR ) 

      END 



Note that t must be consistent between calls to ?geqr and ?gemqr; that is, the t array used in the 

?gemqr call should be the output t array from the previous call to ?geqr, and should not be destroyed 

or overwritten between the calls. 

The following charts show the speedup of DGEQR compared to DGEQRF. The first chart shows these 

speedups on an Intel® Xeon® CPU E5-2699 v4 processor, and the second on an Intel® Xeon Phi™ 7250 

processor.  

M = (100000, 200000, ... , 2000000), N = 100

QR Performance Improvement
Intel® Xeon® CPU E5-2699 v4

QR speedup

All results for M x N matrices, double precision
QR performance improvement is DGEQR vs DGEQRF

Performance tests and ratings are measured using specific computer systems and/or components and 
reflect the approximate performance of Intel products as measured by those tests. Any difference in 
system hardware or software design or configuration may affect actual performance. Buyers should 
consult other sources of information to evaluate the performance of systems or components they are 
considering purchasing. For more information on performance tests and on the performance of Intel 
products, refer to www.intel.com/performance/resources/benchmark_limitations.htm.

www.intel.com/performance/resources/benchmark_limitations.htm

All results for M x N matrices, double precision
QR performance improvement is DGEQR vs DGEQRF

Performance tests and ratings are measured using specific computer systems and/or components and 
reflect the approximate performance of Intel products as measured by those tests. Any difference in 
system hardware or software design or configuration may affect actual performance. Buyers should 
consult other sources of information to evaluate the performance of systems or components they are 
considering purchasing. For more information on performance tests and on the performance of Intel 
products, refer to www.intel.com/performance/resources/benchmark_limitations.htm.

www.intel.com/performance/resources/benchmark_limitations.htm

